Rules for integrands of the form $(g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n$

1. $\int (g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0$

   1. $\int \text{Sec}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0$

      1. $\int \text{Sec}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m+n\in\mathbb{Z}^-$

         1: $\int \text{Sec}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m+n+1==0\,\land\,m\neq-\frac{1}{2}$

Rule: If $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m+n+1==0\,\land\,m\neq-\frac{1}{2}$, then

$$\int \text{Sec}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx\;\rightarrow\;-\frac{b\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n}{a\,f\,(2\,m+1)}$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  b*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(c+d*Csc[e+f*x])^n/(a*f*(2*m+1)) /;
FreeQ[{a,b,c,d,e,f,m,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && EqQ[m+n+1,0] && NeQ[2*m+1,0]
```

2: $\int \sec[e+f x]\,(a+b\sec[e+f x])^m\,(c+d\sec[e+f x])^n\,dx$ when $b c+a d=0\land a^2-b^2=0\land m+n+1\in\mathbb{Z}^-\land m\neq-\frac{1}{2}$

Note: If $n+\frac{1}{2}\in\mathbb{Z}^+\land n+\frac{1}{2}<-(m+n)$, then it is better to drive n to $\frac{1}{2}$ in $n-\frac{1}{2}$ steps.

Rule: If $b c+a d=0\land a^2-b^2=0\land m+n+1\in\mathbb{Z}^-\land m\neq-\frac{1}{2}$, then

$$\int \sec[e+f x]\,(a+b\sec[e+f x])^m\,(c+d\sec[e+f x])^n\,dx\longrightarrow$$

$$-\frac{b\tan[e+f x]\,(a+b\sec[e+f x])^m\,(c+d\sec[e+f x])^n}{a f\,(2m+1)}+\frac{(m+n+1)}{a\,(2m+1)}\int \sec[e+f x]\,(a+b\sec[e+f x])^{m+1}\,(c+d\sec[e+f x])^n\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  b*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(c+d*Csc[e+f*x])^n/(a*f*(2*m+1)) +
  (m+n+1)/(a*(2*m+1))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(c+d*Csc[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && ILtQ[m+n+1,0] && NeQ[2*m+1,0] && Not[LtQ[n,0]] &&
  Not[IGtQ[n+1/2,0] && LtQ[n+1/2,-(m+n)]]
```

2. $\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,(c+d\,\text{Sec}[e+f x])^n\,dx$ when $b c+a d==0 \land a^2-b^2==0 \land m+\frac{1}{2}\in\mathbb{Z}^+$

1. $\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,\sqrt{c+d\,\text{Sec}[e+f x]}\,dx$ when $b c+a d==0 \land a^2-b^2==0$

1: $\int \frac{\text{Sec}[e+f x]\,\sqrt{c+d\,\text{Sec}[e+f x]}}{\sqrt{a+b\,\text{Sec}[e+f x]}}\,dx$ when $b c+a d==0 \land a^2-b^2==0$

Rule: If $b c+a d==0 \land a^2-b^2==0$, then

$$\int \frac{\text{Sec}[e+f x]\,\sqrt{c+d\,\text{Sec}[e+f x]}}{\sqrt{a+b\,\text{Sec}[e+f x]}}\,dx \longrightarrow -\frac{a\,c\,\text{Log}\left[1+\frac{b}{a}\,\text{Sec}[e+f x]\right]\,\text{Tan}[e+f x]}{b\,f\,\sqrt{a+b\,\text{Sec}[e+f x]}\,\sqrt{c+d\,\text{Sec}[e+f x]}}$$

Program code:

```
Int[csc[e_.+f_.*x_]*Sqrt[c_+d_.*csc[e_.+f_.*x_]]/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  a*c*Log[1+b/a*Csc[e+f*x]]*Cot[e+f*x]/(b*f*Sqrt[a+b*Csc[e+f*x]]*Sqrt[c+d*Csc[e+f*x]]) /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0]
```

2: $\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,\sqrt{c+d\,\text{Sec}[e+f x]}\,dx$ when $b c+a d==0 \land a^2-b^2==0 \land m\neq-\frac{1}{2}$

Rule: If $b c+a d==0 \land a^2-b^2==0 \land m\neq-\frac{1}{2}$, then

$$\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,\sqrt{c+d\,\text{Sec}[e+f x]}\,dx \longrightarrow -\frac{2\,a\,c\,\text{Tan}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m}{b\,f\,(2\,m+1)\,\sqrt{c+d\,\text{Sec}[e+f x]}}$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_.*Sqrt[c_+d_.*csc[e_.+f_.*x_]],x_Symbol] :=
  2*a*c*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(b*f*(2*m+1)*Sqrt[c+d*Csc[e+f*x]]) /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && NeQ[m,-1/2]
```

2. $\int \mathrm{Sec}[e+f x]\,(a+b\,\mathrm{Sec}[e+f x])^m\,(c+d\,\mathrm{Sec}[e+f x])^n\,dx$ when $b c+a d==0\land a^2-b^2==0\land m-\frac{1}{2}\in\mathbb{Z}^+$

1: $\int \mathrm{Sec}[e+f x]\,(a+b\,\mathrm{Sec}[e+f x])^m\,(c+d\,\mathrm{Sec}[e+f x])^n\,dx$ when $b c+a d==0\land a^2-b^2==0\land n-\frac{1}{2}\in\mathbb{Z}^+\land m<-\frac{1}{2}$

- Rule: If $b c+a d==0\land a^2-b^2==0\land n-\frac{1}{2}\in\mathbb{Z}^+\land m<-\frac{1}{2}$, then

$$\int \mathrm{Sec}[e+f x]\,(a+b\,\mathrm{Sec}[e+f x])^m\,(c+d\,\mathrm{Sec}[e+f x])^n\,dx\,\longrightarrow$$

$$-\frac{2\,a\,c\,\mathrm{Tan}[e+f x]\,(a+b\,\mathrm{Sec}[e+f x])^m\,(c+d\,\mathrm{Sec}[e+f x])^{n-1}}{b\,f\,(2\,m+1)}-\frac{d\,(2\,n-1)}{b\,(2\,m+1)}\int \mathrm{Sec}[e+f x]\,(a+b\,\mathrm{Sec}[e+f x])^{m+1}\,(c+d\,\mathrm{Sec}[e+f x])^{n-1}\,dx$$

- Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  2*a*c*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(c+d*Csc[e+f*x])^(n-1)/(b*f*(2*m+1)) -
  d*(2*n-1)/(b*(2*m+1))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(c+d*Csc[e+f*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IGtQ[n-1/2,0] && LtQ[m,-1/2]
```

2: $\int \text{Sec}[e+f x] \, (a+b\,\text{Sec}[e+f x])^m \, (c+d\,\text{Sec}[e+f x])^n \, dx$ when $b c+a d=0 \land a^2-b^2=0 \land n-\frac{1}{2}\in\mathbb{Z}^+ \land m \nless -\frac{1}{2}$

Rule: If $b c+a d=0 \land a^2-b^2=0 \land n-\frac{1}{2}\in\mathbb{Z}^+ \land m \nless -\frac{1}{2}$, then

$$\int \text{Sec}[e+f x] \, (a+b\,\text{Sec}[e+f x])^m \, (c+d\,\text{Sec}[e+f x])^n \, dx \longrightarrow$$

$$\frac{d\,\text{Tan}[e+f x] \, (a+b\,\text{Sec}[e+f x])^m \, (c+d\,\text{Sec}[e+f x])^{n-1}}{f\,(m+n)}+\frac{c\,(2 n-1)}{m+n}\int \text{Sec}[e+f x] \, (a+b\,\text{Sec}[e+f x])^m \, (c+d\,\text{Sec}[e+f x])^{n-1} \, dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_.*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -d*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(c+d*Csc[e+f*x])^(n-1)/(f*(m+n)) +
  c*(2*n-1)/(m+n)*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^m*(c+d*Csc[e+f*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IGtQ[n-1/2,0] && Not[LtQ[m,-1/2]] && Not[IGtQ[m-1/2,0] && LtQ[m,n]]
```

3. $\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,(c+d\,\text{Sec}[e+f x])^n\,dx$ when $b c+a d==0\land a^2-b^2==0\land n\in\mathbb{Z}^+\land m<0$

1: $\int \frac{\text{Sec}[e+f x]\,(c+d\,\text{Sec}[e+f x])^n}{\sqrt{a+b\,\text{Sec}[e+f x]}}\,dx$ when $b c+a d==0\land a^2-b^2==0\land n\in\mathbb{Z}^+$

Rule: If $b c+a d==0\land a^2-b^2==0\land n\in\mathbb{Z}^+$, then

$$\int \frac{\text{Sec}[e+f x]\,(c+d\,\text{Sec}[e+f x])^n}{\sqrt{a+b\,\text{Sec}[e+f x]}}\,dx \longrightarrow$$

$$\frac{2\,d\,\text{Tan}[e+f x]\,(c+d\,\text{Sec}[e+f x])^{n-1}}{f\,(2 n-1)\,\sqrt{a+b\,\text{Sec}[e+f x]}}+\frac{2\,c\,(2 n-1)}{2 n-1}\int \frac{\text{Sec}[e+f x]\,(c+d\,\text{Sec}[e+f x])^{n-1}}{\sqrt{a+b\,\text{Sec}[e+f x]}}\,dx$$

–

Program code:

```
Int[csc[e_.+f_.*x_]*(c_+d_.*csc[e_.+f_.*x_])^n_./Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  -2*d*Cot[e+f*x]*(c+d*Csc[e+f*x])^(n-1)/(f*(2*n-1)*Sqrt[a+b*Csc[e+f*x]]) +
  2*c*(2*n-1)/(2*n-1)*Int[Csc[e+f*x]*(c+d*Csc[e+f*x])^(n-1)/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IGtQ[n,0]
```

2: $\int \text{Sec}[e+f x] \, (a+b\,\text{Sec}[e+f x])^m \, (c+d\,\text{Sec}[e+f x])^n \, dx$ when $b c+a d==0 \land a^2-b^2==0 \land n\in\mathbb{Z}^+ \land m<-\frac{1}{2}$

Rule: If $b c+a d==0 \land a^2-b^2==0 \land n\in\mathbb{Z}^+ \land m<-\frac{1}{2}$, then

$$\int \text{Sec}[e+f x] \, (a+b\,\text{Sec}[e+f x])^m \, (c+d\,\text{Sec}[e+f x])^n \, dx \longrightarrow$$

$$-\frac{2 a c \,\text{Tan}[e+f x] \, (a+b\,\text{Sec}[e+f x])^m \, (c+d\,\text{Sec}[e+f x])^{n-1}}{b f \, (2 m+1)}-\frac{d \, (2 n-1)}{b \, (2 m+1)} \int \text{Sec}[e+f x] \, (a+b\,\text{Sec}[e+f x])^{m+1} \, (c+d\,\text{Sec}[e+f x])^{n-1} \, dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  2*a*c*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(c+d*Csc[e+f*x])^(n-1)/(b*f*(2*m+1)) -
  d*(2*n-1)/(b*(2*m+1))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(c+d*Csc[e+f*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IGtQ[n,0] && LtQ[m,-1/2] && IntegerQ[2*m]
```

4: $\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,(c+d\,\text{Sec}[e+f x])^n\,dx$ when $b c+a d=0\land a^2-b^2=0\land m\in\mathbb{Z}\land n\in\mathbb{Z}\land n-m\geq 0\land m n>0$

Derivation: Algebraic simplification

Basis: If $b c+a d=0\land a^2-b^2=0$, then $(a+b\,\text{Sec}[z])\,(c+d\,\text{Sec}[z])=-a c\,\text{Tan}[z]^2$

Rule: If $b c+a d=0\land a^2-b^2=0\land m\in\mathbb{Z}\land n\in\mathbb{Z}\land n-m\geq 0\land m n>0$, then

$$\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,(c+d\,\text{Sec}[e+f x])^n\,dx\longrightarrow(-a c)^m\int(g\,\text{Sec}[e+f x])^p\,\text{Tan}[e+f x]^{2m}\,(c+d\,\text{Sec}[e+f x])^{n-m}\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_.*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  (-a*c)^m*Int[ExpandTrig[csc[e+f*x]*cot[e+f*x]^(2*m),(c+d*csc[e+f*x])^(n-m),x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegersQ[m,n] && GeQ[n-m,0] && GtQ[m*n,0]
```

5: $\int \operatorname{Sec}[e+f x]\,(a+b \operatorname{Sec}[e+f x])^m\,(c+d \operatorname{Sec}[e+f x])^m\,dx$ when $b c+a d==0 \land a^2-b^2==0 \land m-\frac{1}{2}\in \mathbb{Z}^-$

Derivation: Algebraic expansion and piecewise constant extraction

Basis: If $b c+a d==0 \land a^2-b^2==0 \land m+\frac{1}{2}\in \mathbb{Z}$, then $(a+b \operatorname{Sec}[z])^m\,(c+d \operatorname{Sec}[z])^m==\frac{(-a c)^{m+\frac{1}{2}} \operatorname{Tan}[z]^{2 m+1}}{\sqrt{a+b \operatorname{Sec}[z]}\,\sqrt{c+d \operatorname{Sec}[z]}}$

Basis: If $b c+a d==0 \land a^2-b^2==0$, then $\partial_x \frac{\operatorname{Tan}[e+f x]}{\sqrt{a+b \operatorname{Sec}[e+f x]}\,\sqrt{c+d \operatorname{Sec}[e+f x]}}==0$

Rule: If $b c+a d==0 \land a^2-b^2==0 \land m+\frac{1}{2}\in \mathbb{Z}$, then

$$\int \operatorname{Sec}[e+f x]\,(a+b \operatorname{Sec}[e+f x])^m\,(c+d \operatorname{Sec}[e+f x])^m\,dx \longrightarrow$$

$$\frac{(-a c)^{m+\frac{1}{2}} \operatorname{Tan}[e+f x]}{\sqrt{a+b \operatorname{Sec}[e+f x]}\,\sqrt{c+d \operatorname{Sec}[e+f x]}}\int \operatorname{Sec}[e+f x] \operatorname{Tan}[e+f x]^{2 m}\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
  (-a*c)^(m+1/2)*Cot[e+f*x]/(Sqrt[a+b*Csc[e+f*x]]*Sqrt[c+d*Csc[e+f*x]])*Int[Csc[e+f*x]*Cot[e+f*x]^(2*m),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegerQ[m+1/2]
```

6: $\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,(c+d\,\text{Sec}[e+f x])^n\,dx$ when $b c+a d==0 \land a^2-b^2==0 \land \left(\left(m \mid n-\frac{1}{2}\right)\in\mathbb{Z}^- \lor \left(m-\frac{1}{2} \mid n-\frac{1}{2}\right)\in\mathbb{Z}^-\right)$

- Rule: If $b c+a d==0 \land a^2-b^2==0 \land \left(m\in\mathbb{Z}^- \lor \left(m-\frac{1}{2} \mid n-\frac{1}{2}\right)\in\mathbb{Z}^-\right)$, then

$$\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,(c+d\,\text{Sec}[e+f x])^n\,dx \longrightarrow$$

$$-\frac{b\,\text{Tan}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,(c+d\,\text{Sec}[e+f x])^n}{a f\,(2 m+1)}+\frac{(m+n+1)}{a\,(2 m+1)}\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^{m+1}\,(c+d\,\text{Sec}[e+f x])^n\,dx$$

- Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  b*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(c+d*Csc[e+f*x])^n/(a*f*(2*m+1)) +
  (m+n+1)/(a*(2*m+1))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(c+d*Csc[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && (ILtQ[m,0] && ILtQ[n-1/2,0] || ILtQ[m-1/2,0] && ILtQ[n-1/2,0] && LtQ[m,n])
```

7: $\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,(c+d\,\text{Sec}[e+f x])^n\,dx$ when $b c+a d==0 \land a^2-b^2==0$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $b c+a d==0 \land a^2-b^2==0$, then $\partial_x \frac{\text{Tan}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}\,\sqrt{c+d\,\text{Sec}[e+f x]}}==0$

Basis: If $b c+a d==0 \land a^2-b^2==0$, then $-\frac{a c\,\text{Tan}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}\,\sqrt{c+d\,\text{Sec}[e+f x]}}\,\frac{\text{Tan}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}\,\sqrt{c+d\,\text{Sec}[e+f x]}}==1$

Basis: $\text{Tan}[e+f x]\,F[\text{Sec}[e+f x]]==\frac{1}{f}\,\text{Subst}\left[\frac{F[x]}{x},\,x,\,\text{Sec}[e+f x]\right]\partial_x \text{Sec}[e+f x]$

- Rule: If $b c+a d==0 \land a^2-b^2==0$, then

$$\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,(c+d\,\text{Sec}[e+f x])^n\,dx \longrightarrow$$

$$-\frac{a\,c\,\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sec}[e+f\,x]}}\int \mathrm{Tan}[e+f\,x]\,\mathrm{Sec}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m-\frac{1}{2}}\,(c+d\,\mathrm{Sec}[e+f\,x])^{n-\frac{1}{2}}\,dx\longrightarrow$$

$$-\frac{a\,c\,\mathrm{Tan}[e+f\,x]}{f\,\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sec}[e+f\,x]}}\,\mathrm{Subst}\left[\int (a+b\,x)^{m-\frac{1}{2}}\,(c+d\,x)^{n-\frac{1}{2}}\,dx,\,x,\,\mathrm{Sec}[e+f\,x]\right]$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_.*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  a*c*Cot[e+f*x]/(f*Sqrt[a+b*Csc[e+f*x]]*Sqrt[c+d*Csc[e+f*x]])*Subst[Int[(a+b*x)^(m-1/2)*(c+d*x)^(n-1/2),x],x,Csc[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0]
```

2: $\int (g\,\mathrm{Sec}[e+f\,x])^p\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\wedge a^2-b^2==0\wedge m\in\mathbb{Z}\wedge n\in\mathbb{Z}\wedge n-m\geq 0\wedge m\,n>0$

Derivation: Algebraic simplification

Basis: If $b\,c+a\,d==0\wedge a^2-b^2==0$, then $(a+b\,\mathrm{Sec}[z])\,(c+d\,\mathrm{Sec}[z])==-a\,c\,\mathrm{Tan}[z]^2$

Rule: If $b\,c+a\,d==0\wedge a^2-b^2==0\wedge m\in\mathbb{Z}\wedge n\in\mathbb{Z}\wedge n-m\geq 0\wedge m\,n>0$, then

$$\int (g\,\mathrm{Sec}[e+f\,x])^p\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,dx\longrightarrow(-a\,c)^m\int (g\,\mathrm{Sec}[e+f\,x])^p\,\mathrm{Tan}[e+f\,x]^{2m}\,(c+d\,\mathrm{Sec}[e+f\,x])^{n-m}\,dx$$

Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^p_.*(a_+b_.*csc[e_.+f_.*x_])^m_.*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  (-a*c)^m*Int[ExpandTrig[(g*csc[e+f*x])^p*cot[e+f*x]^(2*m),(c+d*csc[e+f*x])^(n-m),x],x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegersQ[m,n] && GeQ[n-m,0] && GtQ[m*n,0]
```

3: $\int (g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^m\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m-\frac{1}{2}\in\mathbb{Z}^-$

Derivation: Algebraic expansion and piecewise constant extraction

Basis: If $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m+\frac{1}{2}\in\mathbb{Z}$, then $(a+b\,\text{Sec}[z])^m\,(c+d\,\text{Sec}[z])^m==\frac{(-a\,c)^{m+\frac{1}{2}}\,\text{Tan}[z]^{2\,m+1}}{\sqrt{a+b\,\text{Sec}[z]}\,\sqrt{c+d\,\text{Sec}[z]}}$

Basis: If $b\,c+a\,d==0\,\land\,a^2-b^2==0$, then $\partial_x\frac{\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}==0$

Rule: If $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m+\frac{1}{2}\in\mathbb{Z}$, then

$$\int (g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^m\,dx\,\longrightarrow$$

$$\frac{(-a\,c)^{m+\frac{1}{2}}\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}\int (g\,\text{Sec}[e+f\,x])^p\,\text{Tan}[e+f\,x]^{2\,m}\,dx$$

Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^p_.*(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
  (-a*c)^(m+1/2)*Cot[e+f*x]/(Sqrt[a+b*Csc[e+f*x]]*Sqrt[c+d*Csc[e+f*x]])*Int[(g*Csc[e+f*x])^p*Cot[e+f*x]^(2*m),x] /;
FreeQ[{a,b,c,d,e,f,g,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegerQ[m+1/2]
```

4: $\int (g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\wedge\,a^2-b^2==0$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $b\,c+a\,d==0\,\wedge\,a^2-b^2==0$, then $\partial_x\frac{\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}==0$

Basis: If $b\,c+a\,d==0\,\wedge\,a^2-b^2==0$, then $-\frac{a\,c\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,\frac{\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}==1$

Basis: $\text{Tan}[e+f\,x]\,F[\text{Sec}[e+f\,x]]==\frac{1}{f}\,\text{Subst}\left[\frac{F[x]}{x},\,x,\,\text{Sec}[e+f\,x]\right]\,\partial_x\text{Sec}[e+f\,x]$

Rule: If $b\,c+a\,d==0\,\wedge\,a^2-b^2==0$, then

$$\int (g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx\,\longrightarrow$$

$$-\frac{a\,c\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}\int \text{Tan}[e+f\,x]\,(g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^{m-\frac{1}{2}}\,(c+d\,\text{Sec}[e+f\,x])^{n-\frac{1}{2}}\,dx\,\longrightarrow$$

$$-\frac{a\,c\,g\,\text{Tan}[e+f\,x]}{f\,\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,\text{Subst}\left[\int (g\,x)^{p-1}\,(a+b\,x)^{m-\frac{1}{2}}\,(c+d\,x)^{n-\frac{1}{2}}\,dx,\,x,\,\text{Sec}[e+f\,x]\right]$$

–

Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^p_.*(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  a*c*g*Cot[e+f*x]/(f*Sqrt[a+b*Csc[e+f*x]]*Sqrt[c+d*Csc[e+f*x]])*
    Subst[Int[(g*x)^(p-1)*(a+b*x)^(m-1/2)*(c+d*x)^(n-1/2),x],x,Csc[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0]
```

2. $\int \frac{(g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^m}{c+d\,\text{Sec}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0$

1. $\int \frac{(g \operatorname{Sec}[e+f x])^p \sqrt{a+b \operatorname{Sec}[e+f x]}}{c+d \operatorname{Sec}[e+f x]} \, dx$ when $b c-a d \neq 0$

1. $\int \frac{\sqrt{g \operatorname{Sec}[e+f x]} \sqrt{a+b \operatorname{Sec}[e+f x]}}{c+d \operatorname{Sec}[e+f x]} \, dx$ when $b c-a d \neq 0$

1: $\int \frac{\sqrt{g \operatorname{Sec}[e+f x]} \sqrt{a+b \operatorname{Sec}[e+f x]}}{c+d \operatorname{Sec}[e+f x]} \, dx$ when $b c-a d \neq 0 \land a^2-b^2 == 0$

Derivation: Integration by substitution

Basis: If $a^2-b^2 == 0$, then $\frac{\sqrt{g \operatorname{Sec}[e+f x]} \sqrt{a+b \operatorname{Sec}[e+f x]}}{c+d \operatorname{Sec}[e+f x]} == \frac{2 b g}{f} \operatorname{Subst}\left[\frac{1}{b c+a d-c g x^2}, x, \frac{b \operatorname{Tan}[e+f x]}{\sqrt{g \operatorname{Sec}[e+f x]} \sqrt{a+b \operatorname{Sec}[e+f x]}}\right] \partial_x \frac{b \operatorname{Tan}[e+f x]}{\sqrt{g \operatorname{Sec}[e+f x]} \sqrt{a+b \operatorname{Sec}[e+f x]}}$

Rule: If $b c-a d \neq 0 \land a^2-b^2 == 0$, then

$$\int \frac{\sqrt{g \operatorname{Sec}[e+f x]} \sqrt{a+b \operatorname{Sec}[e+f x]}}{c+d \operatorname{Sec}[e+f x]} \, dx \rightarrow \frac{2 b g}{f} \operatorname{Subst}\left[\int \frac{1}{b c+a d-c g x^2} \, dx, x, \frac{b \operatorname{Tan}[e+f x]}{\sqrt{g \operatorname{Sec}[e+f x]} \sqrt{a+b \operatorname{Sec}[e+f x]}}\right]$$

Program code:

```
Int[Sqrt[g_.*csc[e_.+f_.*x_]]*Sqrt[a_+b_.*csc[e_.+f_.*x_]]/(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -2*b*g/f*Subst[Int[1/(b*c+a*d-c*g*x^2),x],x,b*Cot[e+f*x]/(Sqrt[g*Csc[e+f*x]]*Sqrt[a+b*Csc[e+f*x]])] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0]
```

2: $\int \frac{\sqrt{g \operatorname{Sec}[e+f x]} \sqrt{a+b \operatorname{Sec}[e+f x]}}{c+d \operatorname{Sec}[e+f x]} \, dx$ when $b c-a d \neq 0 \land a^2-b^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{a+b z}}{c+d z} == \frac{a}{c \sqrt{a+b z}} + \frac{(b c-a d) g z}{c g \sqrt{a+b z} (c+d z)}$

Rule: If $b c-a d \neq 0 \land a^2-b^2 \neq 0$, then

$$\int \frac{\sqrt{g\,\text{Sec}[e+f\,x]}\,\sqrt{a+b\,\text{Sec}[e+f\,x]}}{c+d\,\text{Sec}[e+f\,x]}\,dx \to \frac{a}{c}\int \frac{\sqrt{g\,\text{Sec}[e+f\,x]}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx + \frac{b\,c-a\,d}{c\,g}\int \frac{(g\,\text{Sec}[e+f\,x])^{3/2}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,(c+d\,\text{Sec}[e+f\,x])}\,dx$$

Program code:

```
Int[Sqrt[g_.*csc[e_.+f_.*x_]]*Sqrt[a_+b_.*csc[e_.+f_.*x_]]/(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  a/c*Int[Sqrt[g*Csc[e+f*x]]/Sqrt[a+b*Csc[e+f*x]],x] +
  (b*c-a*d)/(c*g)*Int[(g*Csc[e+f*x])^(3/2)/(Sqrt[a+b*Csc[e+f*x]]*(c+d*Csc[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0]
```

2. $\int \frac{\text{Sec}[e+f\,x]\,\sqrt{a+b\,\text{Sec}[e+f\,x]}}{c+d\,\text{Sec}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{\text{Sec}[e+f\,x]\,\sqrt{a+b\,\text{Sec}[e+f\,x]}}{c+d\,\text{Sec}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\land a^2-b^2==0$

Derivation: Integration by substitution

Basis: If $a^2-b^2==0$, then $\frac{\text{Sec}[e+f\,x]\,\sqrt{a+b\,\text{Sec}[e+f\,x]}}{c+d\,\text{Sec}[e+f\,x]}==\frac{2\,b}{f}\,\text{Subst}\left[\frac{1}{b\,c+a\,d+d\,x^2},\,x,\,\frac{b\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\right]\partial_x\frac{b\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}$

Rule: If $b\,c-a\,d\neq 0\land a^2-b^2==0$, then

$$\int \frac{\text{Sec}[e+f\,x]\,\sqrt{a+b\,\text{Sec}[e+f\,x]}}{c+d\,\text{Sec}[e+f\,x]}\,dx \to \frac{2\,b}{f}\,\text{Subst}\left[\int \frac{1}{b\,c+a\,d+d\,x^2}\,dx,\,x,\,\frac{b\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\right]$$

Program code:

```
Int[csc[e_.+f_.*x_]*Sqrt[a_+b_.*csc[e_.+f_.*x_]]/(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -2*b/f*Subst[Int[1/(b*c+a*d+d*x^2),x],x,b*Cot[e+f*x]/Sqrt[a+b*Csc[e+f*x]]] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0]
```

2. $\int \frac{\text{Sec}[e+f x] \sqrt{a+b\,\text{Sec}[e+f x]}}{c+d\,\text{Sec}[e+f x]}\,dx$ when $b c-a d\neq 0 \land a^2-b^2\neq 0$

1: $\int \frac{\text{Sec}[e+f x] \sqrt{a+b\,\text{Sec}[e+f x]}}{c+d\,\text{Sec}[e+f x]}\,dx$ when $b c-a d\neq 0 \land a^2-b^2\neq 0 \land c^2-d^2==0$

– Rule: If $b c-a d\neq 0 \land a^2-b^2\neq 0 \land c^2-d^2==0$, then

$$\int \frac{\text{Sec}[e+f x] \sqrt{a+b\,\text{Sec}[e+f x]}}{c+d\,\text{Sec}[e+f x]}\,dx \to \frac{\sqrt{a+b\,\text{Sec}[e+f x]}\,\sqrt{\frac{c}{c+d\,\text{Sec}[e+f x]}}}{d f \sqrt{\frac{c d\,(a+b\,\text{Sec}[e+f x])}{(b c+a d)\,(c+d\,\text{Sec}[e+f x])}}}\,\text{EllipticE}\left[\text{ArcSin}\left[\frac{c\,\text{Tan}[e+f x]}{c+d\,\text{Sec}[e+f x]}\right],-\frac{b c-a d}{b c+a d}\right]$$

– Program code:

```
Int[csc[e_.+f_.*x_]*Sqrt[a_+b_.*csc[e_.+f_.*x_]]/(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -Sqrt[a+b*Csc[e+f*x]]*Sqrt[c/(c+d*Csc[e+f*x])]/(d*f*Sqrt[c*d*(a+b*Csc[e+f*x])/((b*c+a*d)*(c+d*Csc[e+f*x]))])*
    EllipticE[ArcSin[c*Cot[e+f*x]/(c+d*Csc[e+f*x])],-(b*c-a*d)/(b*c+a*d)] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && EqQ[c^2-d^2,0]
```

2: $\int \frac{\text{Sec}[e+f x]\sqrt{a+b\,\text{Sec}[e+f x]}}{c+d\,\text{Sec}[e+f x]}\,dx$ when $b c-a d\neq 0\land a^2-b^2\neq 0\land c^2-d^2\neq 0$

- Derivation: Algebraic expansion

- Basis: $\frac{\sqrt{a+b z}}{c+d z}==\frac{b}{d\sqrt{a+b z}}-\frac{b c-a d}{d\sqrt{a+b z}\,(c+d z)}$

- Rule: If $b c-a d\neq 0\land a^2-b^2\neq 0\land c^2-d^2\neq 0$, then

$$\int \frac{\text{Sec}[e+f x]\sqrt{a+b\,\text{Sec}[e+f x]}}{c+d\,\text{Sec}[e+f x]}\,dx\to\frac{b}{d}\int \frac{\text{Sec}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}}\,dx-\frac{b c-a d}{d}\int \frac{\text{Sec}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}\,(c+d\,\text{Sec}[e+f x])}\,dx$$

- Program code:

```
Int[csc[e_.+f_.*x_]*Sqrt[a_+b_.*csc[e_.+f_.*x_]]/(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  b/d*Int[Csc[e+f*x]/Sqrt[a+b*Csc[e+f*x]],x] -
  (b*c-a*d)/d*Int[Csc[e+f*x]/(Sqrt[a+b*Csc[e+f*x]]*(c+d*Csc[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

3. $\int \frac{(g\,\text{Sec}[e+f x])^{3/2}\sqrt{a+b\,\text{Sec}[e+f x]}}{c+d\,\text{Sec}[e+f x]}\,dx$ when $b c-a d\neq 0$

1: $\int \frac{(g\,\text{Sec}[e+f x])^{3/2}\sqrt{a+b\,\text{Sec}[e+f x]}}{c+d\,\text{Sec}[e+f x]}\,dx$ when $b c-a d\neq 0\land a^2-b^2==0$

- Derivation: Algebraic expansion

- Basis: $\frac{(g z)^{3/2}}{c+d z}==\frac{g\sqrt{g z}}{d}-\frac{c g\sqrt{g z}}{d\,(c+d z)}$

- Rule: If $b c-a d\neq 0\land a^2-b^2==0$, then

$$\int \frac{(g \operatorname{Sec}[e+f x])^{3/2} \sqrt{a+b \operatorname{Sec}[e+f x]}}{c+d \operatorname{Sec}[e+f x]} \, dx \to \frac{g}{d} \int \sqrt{g \operatorname{Sec}[e+f x]} \sqrt{a+b \operatorname{Sec}[e+f x]} \, dx - \frac{c g}{d} \int \frac{\sqrt{g \operatorname{Sec}[e+f x]} \sqrt{a+b \operatorname{Sec}[e+f x]}}{c+d \operatorname{Sec}[e+f x]} \, dx$$

Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^(3/2)*Sqrt[a_+b_.*csc[e_.+f_.*x_]]/(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  g/d*Int[Sqrt[g*Csc[e+f*x]]*Sqrt[a+b*Csc[e+f*x]],x] -
  c*g/d*Int[Sqrt[g*Csc[e+f*x]]*Sqrt[a+b*Csc[e+f*x]]/(c+d*Csc[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0]
```

2: $\int \frac{(g \operatorname{Sec}[e+f x])^{3/2} \sqrt{a+b \operatorname{Sec}[e+f x]}}{c+d \operatorname{Sec}[e+f x]} \, dx$ when $b c - a d \neq 0 \wedge a^2 - b^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{a+b z}}{c+d z} == \frac{b}{d \sqrt{a+b z}} - \frac{b c - a d}{d \sqrt{a+b z} \, (c+d z)}$

Rule: If $b c - a d \neq 0 \wedge a^2 - b^2 \neq 0$, then

$$\int \frac{(g \operatorname{Sec}[e+f x])^{3/2} \sqrt{a+b \operatorname{Sec}[e+f x]}}{c+d \operatorname{Sec}[e+f x]} \, dx \to \frac{b}{d} \int \frac{(g \operatorname{Sec}[e+f x])^{3/2}}{\sqrt{a+b \operatorname{Sec}[e+f x]}} \, dx - \frac{b c - a d}{d} \int \frac{(g \operatorname{Sec}[e+f x])^{3/2}}{\sqrt{a+b \operatorname{Sec}[e+f x]} \, (c+d \operatorname{Sec}[e+f x])} \, dx$$

Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^(3/2)*Sqrt[a_+b_.*csc[e_.+f_.*x_]]/(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  b/d*Int[(g*Csc[e+f*x])^(3/2)/Sqrt[a+b*Csc[e+f*x]],x] -
  (b*c-a*d)/d*Int[(g*Csc[e+f*x])^(3/2)/(Sqrt[a+b*Csc[e+f*x]]*(c+d*Csc[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0]
```

2. $\int \frac{(g\,\mathrm{Sec}[e+f\,x])^p}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\;(c+d\,\mathrm{Sec}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0$

1. $\int \frac{\mathrm{Sec}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\;(c+d\,\mathrm{Sec}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{\mathrm{Sec}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\;(c+d\,\mathrm{Sec}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0\wedge\left(a^2-b^2=0\vee c^2-d^2=0\right)$

Derivation: Algebraic expansion

Basis: $\frac{1}{\sqrt{a+b\,z}\;(c+d\,z)}=\frac{b}{(b\,c-a\,d)\sqrt{a+b\,z}}-\frac{d\sqrt{a+b\,z}}{(b\,c-a\,d)\,(c+d\,z)}$

Rule: If $b\,c-a\,d\neq 0\wedge\left(a^2-b^2=0\vee c^2-d^2=0\right)$, then

$$\int \frac{\mathrm{Sec}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\;(c+d\,\mathrm{Sec}[e+f\,x])}\,dx\to\frac{b}{b\,c-a\,d}\int \frac{\mathrm{Sec}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}\,dx-\frac{d}{b\,c-a\,d}\int \frac{\mathrm{Sec}[e+f\,x]\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}{c+d\,\mathrm{Sec}[e+f\,x]}\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]/(Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(c_+d_.*csc[e_.+f_.*x_])),x_Symbol] :=
  b/(b*c-a*d)*Int[Csc[e+f*x]/Sqrt[a+b*Csc[e+f*x]],x] -
  d/(b*c-a*d)*Int[Csc[e+f*x]*Sqrt[a+b*Csc[e+f*x]]/(c+d*Csc[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && (EqQ[a^2-b^2,0] || EqQ[c^2-d^2,0])
```

2: $\int \frac{\mathrm{Sec}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\;(c+d\,\mathrm{Sec}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0\wedge a^2-b^2\neq 0\wedge c^2-d^2\neq 0$

Rule: If $b\,c-a\,d\neq 0\wedge a^2-b^2\neq 0\wedge c^2-d^2\neq 0$, then

$$\int \frac{\mathrm{Sec}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\;(c+d\,\mathrm{Sec}[e+f\,x])}\,dx\to$$

$$\frac{2\,\mathrm{Tan}[e+f\,x]}{f\,(c+d)\,\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{-\mathrm{Tan}[e+f\,x]^2}}\sqrt{\frac{a+b\,\mathrm{Sec}[e+f\,x]}{a+b}}\,\mathrm{EllipticPi}\Big[\frac{2\,d}{c+d},\,\mathrm{ArcSin}\Big[\frac{\sqrt{1-\mathrm{Sec}[e+f\,x]}}{\sqrt{2}}\Big],\,\frac{2\,b}{a+b}\Big]$$

–

Program code:

```
Int[csc[e_.+f_.*x_]/(Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(c_+d_.*csc[e_.+f_.*x_])),x_Symbol] :=
  -2*Cot[e+f*x]/(f*(c+d)*Sqrt[a+b*Csc[e+f*x]]*Sqrt[-Cot[e+f*x]^2])*Sqrt[(a+b*Csc[e+f*x])/(a+b)]*
    EllipticPi[2*d/(c+d),ArcSin[Sqrt[1-Csc[e+f*x]]/Sqrt[2]],2*b/(a+b)] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

2. $\int\frac{(g\,\mathrm{Sec}[e+f\,x])^{3/2}}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,(c+d\,\mathrm{Sec}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq0$

1: $\int\frac{(g\,\mathrm{Sec}[e+f\,x])^{3/2}}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,(c+d\,\mathrm{Sec}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq0\,\wedge\,a^2-b^2==0$

Derivation: Algebraic expansion

Basis: $\frac{g\,z}{\sqrt{a+b\,z}\,(c+d\,z)}==-\frac{a\,g}{(b\,c-a\,d)\,\sqrt{a+b\,z}}+\frac{c\,g\,\sqrt{a+b\,z}}{(b\,c-a\,d)\,(c+d\,z)}$

–

Rule: If $b\,c-a\,d\neq0\,\wedge\,a^2-b^2==0$, then

$$\int\frac{(g\,\mathrm{Sec}[e+f\,x])^{3/2}}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,(c+d\,\mathrm{Sec}[e+f\,x])}\,dx\,\to\,-\frac{a\,g}{b\,c-a\,d}\int\frac{\sqrt{g\,\mathrm{Sec}[e+f\,x]}}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}\,dx+\frac{c\,g}{b\,c-a\,d}\int\frac{\sqrt{g\,\mathrm{Sec}[e+f\,x]}\,\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}{c+d\,\mathrm{Sec}[e+f\,x]}\,dx$$

–

Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^(3/2)/(Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(c_+d_.*csc[e_.+f_.*x_])),x_Symbol] :=
  -a*g/(b*c-a*d)*Int[Sqrt[g*Csc[e+f*x]]/Sqrt[a+b*Csc[e+f*x]],x] +
  c*g/(b*c-a*d)*Int[Sqrt[g*Csc[e+f*x]]*Sqrt[a+b*Csc[e+f*x]]/(c+d*Csc[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0]
```

2: $\int \frac{(g\,\text{Sec}[e+f\,x])^{3/2}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,(c+d\,\text{Sec}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{g\,\text{Sec}[e+f\,x]}\,\sqrt{b+a\,\text{Cos}[e+f\,x]}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}} == 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0$, then

$$\int \frac{(g\,\text{Sec}[e+f\,x])^{3/2}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,(c+d\,\text{Sec}[e+f\,x])}\,dx \to \frac{g\,\sqrt{g\,\text{Sec}[e+f\,x]}\,\sqrt{b+a\,\text{Cos}[e+f\,x]}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\int \frac{1}{\sqrt{b+a\,\text{Cos}[e+f\,x]}\,(d+c\,\text{Cos}[e+f\,x])}\,dx$$

Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^(3/2)/(Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(c_+d_.*csc[e_.+f_.*x_])),x_Symbol] :=
  g*Sqrt[g*Csc[e+f*x]]*Sqrt[b+a*Sin[e+f*x]]/Sqrt[a+b*Csc[e+f*x]]*Int[1/(Sqrt[b+a*Sin[e+f*x]]*(d+c*Sin[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0]
```

3. $\int \frac{\text{Sec}[e+f\,x]^2}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,(c+d\,\text{Sec}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{\text{Sec}[e+f\,x]^2}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,(c+d\,\text{Sec}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,(a^2-b^2==0\,\vee\,c^2-d^2==0)$

Derivation: Algebraic expansion

Basis: $\frac{z^2}{\sqrt{a+b\,z}\,(c+d\,z)} == -\frac{a\,z}{(b\,c-a\,d)\,\sqrt{a+b\,z}}+\frac{c\,z\,\sqrt{a+b\,z}}{(b\,c-a\,d)\,(c+d\,z)}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,(a^2-b^2==0\,\vee\,c^2-d^2==0)$, then

$$\int \frac{\text{Sec}[e+f x]^2}{\sqrt{a+b\,\text{Sec}[e+f x]}\ (c+d\,\text{Sec}[e+f x])}\,dx \to -\frac{a}{b c-a d}\int \frac{\text{Sec}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}}\,dx+\frac{c}{b c-a d}\int \frac{\text{Sec}[e+f x]\sqrt{a+b\,\text{Sec}[e+f x]}}{c+d\,\text{Sec}[e+f x]}\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^2/(Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(c_+d_.*csc[e_.+f_.*x_])),x_Symbol] :=
  -a/(b*c-a*d)*Int[Csc[e+f*x]/Sqrt[a+b*Csc[e+f*x]],x] +
  c/(b*c-a*d)*Int[Csc[e+f*x]*Sqrt[a+b*Csc[e+f*x]]/(c+d*Csc[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && (EqQ[a^2-b^2,0] || EqQ[c^2-d^2,0])
```

2: $\int \frac{\text{Sec}[e+f x]^2}{\sqrt{a+b\,\text{Sec}[e+f x]}\ (c+d\,\text{Sec}[e+f x])}\,dx$ when $b c-a d\neq 0\land a^2-b^2\neq 0\land c^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{z^2}{\sqrt{a+b z}\ (c+d z)} == \frac{z}{d\sqrt{a+b z}}-\frac{c z}{d\sqrt{a+b z}\ (c+d z)}$

Rule: If $b c-a d\neq 0\land a^2-b^2\neq 0\land c^2-d^2\neq 0$, then

$$\int \frac{\text{Sec}[e+f x]^2}{\sqrt{a+b\,\text{Sec}[e+f x]}\ (c+d\,\text{Sec}[e+f x])}\,dx \to \frac{1}{d}\int \frac{\text{Sec}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}}\,dx-\frac{c}{d}\int \frac{\text{Sec}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}\ (c+d\,\text{Sec}[e+f x])}\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^2/(Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(c_+d_.*csc[e_.+f_.*x_])),x_Symbol] :=
  1/d*Int[Csc[e+f*x]/Sqrt[a+b*Csc[e+f*x]],x] -
  c/d*Int[Csc[e+f*x]/(Sqrt[a+b*Csc[e+f*x]]*(c+d*Csc[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

4. $\int \frac{(g \operatorname{Sec}[e+fx])^{5/2}}{\sqrt{a+b \operatorname{Sec}[e+fx]}\,(c+d \operatorname{Sec}[e+fx])} \, dx$ when $b c - a d \neq 0$

1: $\int \frac{(g \operatorname{Sec}[e+fx])^{5/2}}{\sqrt{a+b \operatorname{Sec}[e+fx]}\,(c+d \operatorname{Sec}[e+fx])} \, dx$ when $b c - a d \neq 0 \wedge a^2 - b^2 == 0$

Derivation: Algebraic expansion

Basis: $\frac{g^2 z^2}{\sqrt{a+b z}\,(c+d z)} == -\frac{c^2 g^2 \sqrt{a+b z}}{d\,(b c-a d)\,(c+d z)} + \frac{g^2\,(a c+(b c-a d)\,z)}{d\,(b c-a d)\,\sqrt{a+b z}}$

Rule: If $b c - a d \neq 0 \wedge a^2 - b^2 == 0$, then

$$\int \frac{(g \operatorname{Sec}[e+fx])^{5/2}}{\sqrt{a+b \operatorname{Sec}[e+fx]}\,(c+d \operatorname{Sec}[e+fx])} \, dx \to -\frac{c^2 g^2}{d\,(b c-a d)} \int \frac{\sqrt{g \operatorname{Sec}[e+fx]}\,\sqrt{a+b \operatorname{Sec}[e+fx]}}{c+d \operatorname{Sec}[e+fx]} \, dx + \frac{g^2}{d\,(b c-a d)} \int \frac{\sqrt{g \operatorname{Sec}[e+fx]}\,(a c+(b c-a d) \operatorname{Sec}[e+fx])}{\sqrt{a+b \operatorname{Sec}[e+fx]}} \, dx$$

Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^(5/2)/(Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(c_+d_.*csc[e_.+f_.*x_])),x_Symbol] :=
  -c^2*g^2/(d*(b*c-a*d))*Int[Sqrt[g*Csc[e+f*x]]*Sqrt[a+b*Csc[e+f*x]]/(c+d*Csc[e+f*x]),x] +
  g^2/(d*(b*c-a*d))*Int[Sqrt[g*Csc[e+f*x]]*(a*c+(b*c-a*d)*Csc[e+f*x])/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0]
```

2: $\int \frac{(g \operatorname{Sec}[e+f x])^{5/2}}{\sqrt{a+b \operatorname{Sec}[e+f x]}\,(c+d \operatorname{Sec}[e+f x])} \, dx$ when $b c-a d \neq 0 \land a^2-b^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{g z}{c+d z} == \frac{g}{d} - \frac{c g}{d (c+d z)}$

Rule: If $b c-a d \neq 0 \land a^2-b^2 \neq 0$, then

$$\int \frac{(g \operatorname{Sec}[e+f x])^{5/2}}{\sqrt{a+b \operatorname{Sec}[e+f x]}\,(c+d \operatorname{Sec}[e+f x])} \, dx \to \frac{g}{d} \int \frac{(g \operatorname{Sec}[e+f x])^{3/2}}{\sqrt{a+b \operatorname{Sec}[e+f x]}} \, dx - \frac{c g}{d} \int \frac{(g \operatorname{Sec}[e+f x])^{3/2}}{\sqrt{a+b \operatorname{Sec}[e+f x]}\,(c+d \operatorname{Sec}[e+f x])} \, dx$$

Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^(5/2)/(Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(c_+d_.*csc[e_.+f_.*x_])),x_Symbol] :=
  g/d*Int[(g*Csc[e+f*x])^(3/2)/Sqrt[a+b*Csc[e+f*x]],x] -
  c*g/d*Int[(g*Csc[e+f*x])^(3/2)/(Sqrt[a+b*Csc[e+f*x]]*(c+d*Csc[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0]
```

3. $\int \frac{\operatorname{Sec}[e+f x]^p (a+b \operatorname{Sec}[e+f x])^m}{\sqrt{c+d \operatorname{Sec}[e+f x]}} \, dx$ when $b c-a d \neq 0 \land m^2 == \frac{1}{4}$

1. $\int \frac{\operatorname{Sec}[e+f x] \sqrt{a+b \operatorname{Sec}[e+f x]}}{\sqrt{c+d \operatorname{Sec}[e+f x]}} \, dx$ when $b c-a d \neq 0$

1: $\int \frac{\operatorname{Sec}[e+f x] \sqrt{a+b \operatorname{Sec}[e+f x]}}{\sqrt{c+d \operatorname{Sec}[e+f x]}} \, dx$ when $b c-a d \neq 0 \land a^2-b^2 == 0 \land c^2-d^2 \neq 0$

Derivation: Integration by substitution

Basis: If $a^2-b^2 == 0 \land c^2-d^2 \neq 0$, then

$$\frac{\text{Sec}[e+f x] \sqrt{a+b\,\text{Sec}[e+f x]}}{\sqrt{c+d\,\text{Sec}[e+f x]}} = \frac{2 b}{f} \text{Subst}\left[\frac{1}{1-b d x^2}, x, \frac{\text{Tan}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]} \sqrt{c+d\,\text{Sec}[e+f x]}}\right] \partial_x \frac{\text{Tan}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]} \sqrt{c+d\,\text{Sec}[e+f x]}}$$

Rule: If $b c - a d \neq 0 \land a^2 - b^2 == 0 \land c^2 - d^2 \neq 0$, then

$$\int \frac{\text{Sec}[e+f x] \sqrt{a+b\,\text{Sec}[e+f x]}}{\sqrt{c+d\,\text{Sec}[e+f x]}} \, dx \to \frac{2 b}{f} \text{Subst}\left[\int \frac{1}{1-b d x^2} \, dx, x, \frac{\text{Tan}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]} \sqrt{c+d\,\text{Sec}[e+f x]}}\right]$$

Program code:

```
Int[csc[e_.+f_.*x_]*Sqrt[a_+b_.*csc[e_.+f_.*x_]]/Sqrt[c_+d_.*csc[e_.+f_.*x_]],x_Symbol] :=
  -2*b/f*Subst[Int[1/(1-b*d*x^2),x],x,Cot[e+f*x]/(Sqrt[a+b*Csc[e+f*x]]*Sqrt[c+d*Csc[e+f*x]])] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

2: $\int \frac{\text{Sec}[e+f x] \sqrt{a+b\,\text{Sec}[e+f x]}}{\sqrt{c+d\,\text{Sec}[e+f x]}} \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 == 0$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{a+b z}}{\sqrt{c+d z}} == -\frac{b c - a d}{d \sqrt{a+b z} \sqrt{c+d z}} + \frac{b \sqrt{c+d z}}{d \sqrt{a+b z}}$

Rule: If $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 == 0$, then

$$\int \frac{\text{Sec}[e+f x] \sqrt{a+b\,\text{Sec}[e+f x]}}{\sqrt{c+d\,\text{Sec}[e+f x]}} \, dx \to -\frac{b c - a d}{d} \int \frac{\text{Sec}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]} \sqrt{c+d\,\text{Sec}[e+f x]}} \, dx + \frac{b}{d} \int \frac{\text{Sec}[e+f x] \sqrt{c+d\,\text{Sec}[e+f x]}}{\sqrt{a+b\,\text{Sec}[e+f x]}} \, dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*Sqrt[a_+b_.*csc[e_.+f_.*x_]]/Sqrt[c_+d_.*csc[e_.+f_.*x_]],x_Symbol] :=
  -(b*c-a*d)/d*Int[Csc[e+f*x]/(Sqrt[a+b*Csc[e+f*x]]*Sqrt[c+d*Csc[e+f*x]]),x] +
  b/d*Int[Csc[e+f*x]*Sqrt[c+d*Csc[e+f*x]]/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && EqQ[c^2-d^2,0]
```

3: $\int \frac{\text{Sec}[e+f x] \sqrt{a+b \text{Sec}[e+f x]}}{\sqrt{c+d \text{Sec}[e+f x]}} \, dx$ when $b c-a d \neq 0 \land a^2-b^2 \neq 0 \land c^2-d^2 \neq 0$

Rule: If $b c-a d \neq 0 \land a^2-b^2 \neq 0 \land c^2-d^2 \neq 0$, then

$$\int \frac{\text{Sec}[e+f x] \sqrt{a+b \text{Sec}[e+f x]}}{\sqrt{c+d \text{Sec}[e+f x]}} \, dx \to$$

$$\frac{2 (a+b \text{Sec}[e+f x])}{d f \sqrt{\frac{a+b}{c+d}} \text{Tan}[e+f x]} \sqrt{-\frac{(b c-a d) (1-\text{Sec}[e+f x])}{(c+d) (a+b \text{Sec}[e+f x])}}$$

$$\sqrt{\frac{(b c-a d) (1+\text{Sec}[e+f x])}{(c-d) (a+b \text{Sec}[e+f x])}} \text{EllipticPi}\left[\frac{b (c+d)}{d (a+b)}, \text{ArcSin}\left[\sqrt{\frac{a+b}{c+d}} \frac{\sqrt{c+d \text{Sec}[e+f x]}}{\sqrt{a+b \text{Sec}[e+f x]}}\right], \frac{(a-b) (c+d)}{(a+b) (c-d)}\right]$$

Program code:

```
Int[csc[e_.+f_.*x_]*Sqrt[a_+b_.*csc[e_.+f_.*x_]]/Sqrt[c_+d_.*csc[e_.+f_.*x_]],x_Symbol] :=
  -2*(a+b*Csc[e+f*x])/(d*f*Sqrt[(a+b)/(c+d)]*Cot[e+f*x])*
    Sqrt[-(b*c-a*d)*(1-Csc[e+f*x])/((c+d)*(a+b*Csc[e+f*x]))]*Sqrt[(b*c-a*d)*(1+Csc[e+f*x])/((c-d)*(a+b*Csc[e+f*x]))]*
    EllipticPi[b*(c+d)/(d*(a+b)),ArcSin[Sqrt[(a+b)/(c+d)]*Sqrt[c+d*Csc[e+f*x]]/Sqrt[a+b*Csc[e+f*x]]],(a-b)*(c+d)/((a+b)*(c-d))] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

2. $\int \frac{\text{Sec}[e+f x]}{\sqrt{a+b \text{Sec}[e+f x]} \sqrt{c+d \text{Sec}[e+f x]}} \, dx$ when $b c-a d \neq 0$

1: $\int \frac{\text{Sec}[e+f x]}{\sqrt{a+b \text{Sec}[e+f x]} \sqrt{c+d \text{Sec}[e+f x]}} \, dx$ when $b c-a d \neq 0 \land a^2-b^2 = 0 \land c^2-d^2 \neq 0$

Derivation: Integration by substitution

Basis: If $a^2 - b^2 == 0 \land c^2 - d^2 \neq 0$, then

$$\frac{\text{Sec}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}\ \sqrt{c+d\,\text{Sec}[e+f x]}} == \frac{2a}{b f}\,\text{Subst}\left[\frac{1}{2+(a c-b d)\,x^2},\ x,\ \frac{\text{Tan}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}\ \sqrt{c+d\,\text{Sec}[e+f x]}}\right] \partial_x \frac{\text{Tan}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}\ \sqrt{c+d\,\text{Sec}[e+f x]}}$$

Rule: If $b c - a d \neq 0 \land a^2 - b^2 == 0 \land c^2 - d^2 \neq 0$, then

$$\int \frac{\text{Sec}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}\ \sqrt{c+d\,\text{Sec}[e+f x]}}\,dx \to \frac{2a}{b f}\,\text{Subst}\left[\int \frac{1}{2+(a c-b d)\,x^2}\,dx,\ x,\ \frac{\text{Tan}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}\ \sqrt{c+d\,\text{Sec}[e+f x]}}\right]$$

Program code:

```
Int[csc[e_.+f_.*x_]/(Sqrt[a_+b_.*csc[e_.+f_.*x_]]*Sqrt[c_+d_.*csc[e_.+f_.*x_]]),x_Symbol] :=
  -2*a/(b*f)*Subst[Int[1/(2+(a*c-b*d)*x^2),x],x,Cot[e+f*x]/(Sqrt[a+b*Csc[e+f*x]]*Sqrt[c+d*Csc[e+f*x]])] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

2: $\int \frac{\text{Sec}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}\,\sqrt{c+d\,\text{Sec}[e+f x]}}\,dx$ when $b c-a d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0$

Rule: If $b c-a d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0$, then

$$\int \frac{\text{Sec}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}\,\sqrt{c+d\,\text{Sec}[e+f x]}}\,dx\,\rightarrow$$

$$\frac{2\,(c+d\,\text{Sec}[e+f x])}{f\,(b c-a d)\,\sqrt{\frac{c+d}{a+b}}\,\text{Tan}[e+f x]}\sqrt{\frac{(b c-a d)\,(1-\text{Sec}[e+f x])}{(a+b)\,(c+d\,\text{Sec}[e+f x])}}$$

$$\sqrt{-\frac{(b c-a d)\,(1+\text{Sec}[e+f x])}{(a-b)\,(c+d\,\text{Sec}[e+f x])}}\;\text{EllipticF}\Big[\text{ArcSin}\Big[\sqrt{\frac{c+d}{a+b}}\,\frac{\sqrt{a+b\,\text{Sec}[e+f x]}}{\sqrt{c+d\,\text{Sec}[e+f x]}}\Big],\,\frac{(a+b)\,(c-d)}{(a-b)\,(c+d)}\Big]$$

Program code:

```
Int[csc[e_.+f_.*x_]/(Sqrt[a_+b_.*csc[e_.+f_.*x_]]*Sqrt[c_+d_.*csc[e_.+f_.*x_]]),x_Symbol] :=
  -2*(c+d*Csc[e+f*x])/(f*(b*c-a*d)*Rt[(c+d)/(a+b),2]*Cot[e+f*x])*
    Sqrt[(b*c-a*d)*(1-Csc[e+f*x])/((a+b)*(c+d*Csc[e+f*x]))]*Sqrt[-(b*c-a*d)*(1+Csc[e+f*x])/((a-b)*(c+d*Csc[e+f*x]))]*
    EllipticF[ArcSin[Rt[(c+d)/(a+b),2]*(Sqrt[a+b*Csc[e+f*x]]/Sqrt[c+d*Csc[e+f*x]])],(a+b)*(c-d)/((a-b)*(c+d))] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

3: $\int \frac{\text{Sec}[e+f x]^2}{\sqrt{a+b\,\text{Sec}[e+f x]}\,\sqrt{c+d\,\text{Sec}[e+f x]}}\,dx$ when $b c-a d\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{z}{\sqrt{a+b z}}==-\frac{a}{b\sqrt{a+b z}}+\frac{\sqrt{a+b z}}{b}$

Rule: If $b c-a d\neq 0$, then

$$\int \frac{\text{Sec}[e+f x]^2}{\sqrt{a+b\,\text{Sec}[e+f x]}\,\sqrt{c+d\,\text{Sec}[e+f x]}}\,dx\to -\frac{a}{b}\int \frac{\text{Sec}[e+f x]}{\sqrt{a+b\,\text{Sec}[e+f x]}\,\sqrt{c+d\,\text{Sec}[e+f x]}}\,dx+\frac{1}{b}\int \frac{\text{Sec}[e+f x]\,\sqrt{a+b\,\text{Sec}[e+f x]}}{\sqrt{c+d\,\text{Sec}[e+f x]}}\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^2/(Sqrt[a_+b_.*csc[e_.+f_.*x_]]*Sqrt[c_+d_.*csc[e_.+f_.*x_]]),x_Symbol] :=
  -a/b*Int[Csc[e+f*x]/(Sqrt[a+b*Csc[e+f*x]]*Sqrt[c+d*Csc[e+f*x]]),x] +
  1/b*Int[Csc[e+f*x]*Sqrt[a+b*Csc[e+f*x]]/Sqrt[c+d*Csc[e+f*x]],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0]
```

4: $\int \frac{\text{Sec}[e+f x]\,\sqrt{a+b\,\text{Sec}[e+f x]}}{(c+d\,\text{Sec}[e+f x])^{3/2}}\,dx$ when $b c-a d\neq 0\land a^2-b^2\neq 0\land c^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{a+b z}}{c+d z}==\frac{a-b}{(c-d)\sqrt{a+b z}}+\frac{(b c-a d)(1+z)}{(c-d)\sqrt{a+b z}\,(c+d z)}$

Rule: If $b c-a d\neq 0\land a^2-b^2\neq 0\land c^2-d^2\neq 0$, then

$$\int \frac{\text{Sec}[e+f x]\,\sqrt{a+b\,\text{Sec}[e+f x]}}{(c+d\,\text{Sec}[e+f x])^{3/2}}\,dx\to$$

$$\frac{a-b}{c-d}\int\frac{\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx+\frac{b\,c-a\,d}{c-d}\int\frac{\text{Sec}[e+f\,x]\,(1+\text{Sec}[e+f\,x])}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,(c+d\,\text{Sec}[e+f\,x])^{3/2}}\,dx$$

–

Program code:

```
Int[csc[e_.+f_.*x_]*Sqrt[a_+b_.*csc[e_.+f_.*x_]]/(c_+d_.*csc[e_.+f_.*x_])^(3/2),x_Symbol] :=
  (a-b)/(c-d)*Int[Csc[e+f*x]/(Sqrt[a+b*Csc[e+f*x]]*Sqrt[c+d*Csc[e+f*x]]),x] +
  (b*c-a*d)/(c-d)*Int[Csc[e+f*x]*(1+Csc[e+f*x])/(Sqrt[a+b*Csc[e+f*x]]*(c+d*Csc[e+f*x])^(3/2)),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

5: $\int (g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,\left(p==1\,\lor\,m-\frac{1}{2}\in\mathbb{Z}\right)$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $a^2-b^2==0$, then $\partial_x \frac{\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{a-b\,\text{Sec}[e+f\,x]}}==0$

Basis: If $a^2-b^2==0$, then $-\frac{a^2\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{a-b\,\text{Sec}[e+f\,x]}}\;\frac{\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{a-b\,\text{Sec}[e+f\,x]}}==1$

Basis: $\text{Tan}[e+f\,x]\,F[\text{Sec}[e+f\,x]]==\frac{1}{f}\,\text{Subst}\left[\frac{F[x]}{x},\,x,\,\text{Sec}[e+f\,x]\right]\,\partial_x\,\text{Sec}[e+f\,x]$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,\left(p==1\,\lor\,m-\frac{1}{2}\in\mathbb{Z}\right)$, then

$$\int (g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx\;\longrightarrow$$

$$-\frac{a^2\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{a-b\,\text{Sec}[e+f\,x]}}\int\frac{\text{Tan}[e+f\,x]\,(g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^{m-\frac{1}{2}}\,(c+d\,\text{Sec}[e+f\,x])^n}{\sqrt{a-b\,\text{Sec}[e+f\,x]}}\,dx\;\longrightarrow$$

$$-\frac{a^2\,g\,\text{Tan}[e+f\,x]}{f\,\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{a-b\,\text{Sec}[e+f\,x]}}\,\text{Subst}\left[\int\frac{(g\,x)^{p-1}\,(a+b\,x)^{m-\frac{1}{2}}\,(c+d\,x)^n}{\sqrt{a-b\,x}}\,dx,\,x,\,\text{Sec}[e+f\,x]\right]$$

Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^p_.*(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  a^2*g*Cot[e+f*x]/(f*Sqrt[a+b*Csc[e+f*x]]*Sqrt[a-b*Csc[e+f*x]])*
    Subst[Int[(g*x)^(p-1)*(a+b*x)^(m-1/2)*(c+d*x)^n/Sqrt[a-b*x],x],x,Csc[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && (EqQ[p,1] || IntegerQ[m-1/2])
```

6: $\int (g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,m\in\mathbb{Z}\,\land\,n\in\mathbb{Z}$

Derivation: Algebraic normalization

Basis: If $m\in\mathbb{Z}\,\land\,n\in\mathbb{Z}$, then $(a+b\,\text{Sec}[z])^m\,(c+d\,\text{Sec}[z])^n == \text{Sec}[z]^{m+n}\,(b+a\,\text{Cos}[z])^m\,(d+c\,\text{Cos}[z])^n$

Rule: If $b\,c-a\,d\neq 0\,\land\,m\in\mathbb{Z}\,\land\,n\in\mathbb{Z}$, then

$$\int (g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx\,\rightarrow\,\frac{1}{g^{m+n}}\int (g\,\text{Sec}[e+f\,x])^{m+n+p}\,(b+a\,\text{Cos}[e+f\,x])^m\,(d+c\,\text{Cos}[e+f\,x])^n\,dx$$

Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^p_.*(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  1/g^(m+n)*Int[(g*Csc[e+f*x])^(m+n+p)*(b+a*Sin[e+f*x])^m*(d+c*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,p},x] && NeQ[b*c-a*d,0] && IntegerQ[m] && IntegerQ[n]
```

7. $\int (g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,m+n+p==0$

1: $\int (g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,m+n+p==0\,\land\,m\in\mathbb{Z}$

Derivation: Algebraic normalization and piecewise constant extraction

Basis: $a+b\,\text{Sec}[e+f\,x] == \text{Sec}[e+f\,x]\,(b+a\,\text{Cos}[e+f\,x])$

Basis: If $m+n+p==0$, then $\partial_x \frac{(g\,\text{Sec}[e+f\,x])^{m+p}\,(c+d\,\text{Sec}[e+f\,x])^n}{(d+c\,\text{Cos}[e+f\,x])^n} == 0$

Rule: If $b\,c-a\,d\neq 0\,\land\,m+n+p==0\,\land\,m\in\mathbb{Z}$, then

$$\int (g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx\,\rightarrow\,\frac{1}{g^m}\int (g\,\text{Sec}[e+f\,x])^{m+p}\,(b+a\,\text{Cos}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx$$

$$\longrightarrow \frac{(g\,\text{Sec}[e+f\,x])^{m+p}\,(c+d\,\text{Sec}[e+f\,x])^n}{g^m\,(d+c\,\text{Cos}[e+f\,x])^n}\int(b+a\,\text{Cos}[e+f\,x])^m\,(d+c\,\text{Cos}[e+f\,x])^n\,dx$$

–

Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^p_.*(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  (g*Csc[e+f*x])^(m+p)*(c+d*Csc[e+f*x])^n/(g^m*(d+c*Sin[e+f*x])^n)*Int[(b+a*Sin[e+f*x])^m*(d+c*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && NeQ[b*c-a*d,0] && EqQ[m+n+p,0] && IntegerQ[m]
```

2: $\int(g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,m+n+p=0\,\wedge\,m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $m+n+p=0$, then $\partial_x\frac{(g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n}{(b+a\,\text{Cos}[e+f\,x])^m\,(d+c\,\text{Cos}[e+f\,x])^n}=0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,m+n+p=0\,\wedge\,m\notin\mathbb{Z}$, then

$$\int(g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx\longrightarrow\frac{(g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n}{(b+a\,\text{Cos}[e+f\,x])^m\,(d+c\,\text{Cos}[e+f\,x])^n}\int(b+a\,\text{Cos}[e+f\,x])^m\,(d+c\,\text{Cos}[e+f\,x])^n\,dx$$

–

Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^p_.*(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  (g*Csc[e+f*x])^p*(a+b*Csc[e+f*x])^m*(c+d*Csc[e+f*x])^n/((b+a*Sin[e+f*x])^m*(d+c*Sin[e+f*x])^n)*
    Int[(b+a*Sin[e+f*x])^m*(d+c*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && NeQ[b*c-a*d,0] && EqQ[m+n+p,0] && Not[IntegerQ[m]]
```

8: $\int \text{Sec}[e+f x]^p (a+b\,\text{Sec}[e+f x])^m (c+d\,\text{Sec}[e+f x])^n\,dx$ when $b c-a d\neq 0 \land m-\frac{1}{2}\in\mathbb{Z} \land n-\frac{1}{2}\in\mathbb{Z} \land p\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{d+c\,\text{Cos}[e+f x]}\,\sqrt{a+b\,\text{Sec}[e+f x]}}{\sqrt{b+a\,\text{Cos}[e+f x]}\,\sqrt{c+d\,\text{Sec}[e+f x]}} == 0$

Note: The restriction $m+n+p\in\{-1,-2\}$ can be lifted if and when the cosine integration rules are extended to handle integrands of the form $\text{Cos}[e+f x]^p (a+b\,\text{Cos}[e+f x])^m (c+d\,\text{Cos}[e+f x])^n$ for arbitray p.

Rule: If $b c-a d\neq 0 \land m-\frac{1}{2}\in\mathbb{Z} \land n-\frac{1}{2}\in\mathbb{Z} \land p\in\mathbb{Z}$, then

$$\int \text{Sec}[e+f x]^p (a+b\,\text{Sec}[e+f x])^m (c+d\,\text{Sec}[e+f x])^n\,dx \to \frac{\sqrt{d+c\,\text{Cos}[e+f x]}\,\sqrt{a+b\,\text{Sec}[e+f x]}}{\sqrt{b+a\,\text{Cos}[e+f x]}\,\sqrt{c+d\,\text{Sec}[e+f x]}}\int \frac{(b+a\,\text{Cos}[e+f x])^m (d+c\,\text{Cos}[e+f x])^n}{\text{Cos}[e+f x]^{m+n+p}}\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^p_.*(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  Sqrt[d+c*Sin[e+f*x]]*Sqrt[a+b*Csc[e+f*x]]/(Sqrt[b+a*Sin[e+f*x]]*Sqrt[c+d*Csc[e+f*x]])*
    Int[(b+a*Sin[e+f*x])^m*(d+c*Sin[e+f*x])^n/Sin[e+f*x]^(m+n+p),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && NeQ[b*c-a*d,0] && IntegerQ[m-1/2] && IntegerQ[n-1/2] && IntegerQ[p] && LeQ[-2,m+n+p,-1]
```

9: $\int (g\,\text{Sec}[e+f x])^p (a+b\,\text{Sec}[e+f x])^m (c+d\,\text{Sec}[e+f x])^n\,dx$ when $b c-a d\neq 0 \land ((m\mid n)\in\mathbb{Z} \lor (m\mid p)\in\mathbb{Z} \lor (n\mid p)\in\mathbb{Z})$

Derivation: Algebraic expansion

Rule: If $b c-a d\neq 0 \land ((m\mid n)\in\mathbb{Z} \lor (m\mid p)\in\mathbb{Z} \lor (n\mid p)\in\mathbb{Z})$, then

$$\int (g\,\text{Sec}[e+f x])^p (a+b\,\text{Sec}[e+f x])^m (c+d\,\text{Sec}[e+f x])^n\,dx \to$$

$$\int \text{ExpandTrig}\left[(g \,\text{Sec}[e+f x])^p (a+b \,\text{Sec}[e+f x])^m (c+d \,\text{Sec}[e+f x])^n, x\right] dx$$

– Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^p_.*(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  Int[ExpandTrig[(g*csc[e+f*x])^p*(a+b*csc[e+f*x])^m*(c+d*csc[e+f*x])^n,x],x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && NeQ[b*c-a*d,0] && (IntegersQ[m,n] || IntegersQ[m,p] || IntegersQ[n,p])
```

X: $\int (g \,\text{Sec}[e+f x])^p (a+b \,\text{Sec}[e+f x])^m (c+d \,\text{Sec}[e+f x])^n dx$

– Rule:

$$\int (g \,\text{Sec}[e+f x])^p (a+b \,\text{Sec}[e+f x])^m (c+d \,\text{Sec}[e+f x])^n dx \longrightarrow \int (g \,\text{Sec}[e+f x])^p (a+b \,\text{Sec}[e+f x])^m (c+d \,\text{Sec}[e+f x])^n dx$$

– Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^p_.*(a_.+b_.*csc[e_.+f_.*x_])^m_*(c_.+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  Unintegrable[(g*Csc[e+f*x])^p*(a+b*Csc[e+f*x])^m*(c+d*Csc[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x]
```

Rules for integrands of the form $(g \,\text{Sec}[e+f x])^p (a+b \,\text{Sec}[e+f x])^m (c+d \,\text{Sec}[e+f x])^n (A+B \,\text{Sec}[e+f x])$

1: $\int \frac{\text{Sec}[e+f x] (A+B \,\text{Sec}[e+f x])}{\sqrt{a+b \,\text{Sec}[e+f x]} (c+d \,\text{Sec}[e+f x])^{3/2}} dx$ when $b c-a d \neq 0 \land a^2-b^2 \neq 0 \land c^2-d^2 \neq 0 \land A == B$

– Rule: If $b c-a d \neq 0 \land a^2-b^2 \neq 0 \land c^2-d^2 \neq 0 \land A == B$, then

$$\int \frac{\text{Sec}[e+f x] (A+B \,\text{Sec}[e+f x])}{\sqrt{a+b \,\text{Sec}[e+f x]} (c+d \,\text{Sec}[e+f x])^{3/2}} dx \longrightarrow$$

$$\frac{2 A\,(1+\operatorname{Sec}[e+f x])\,\sqrt{\frac{(b c-a d)\,(1-\operatorname{Sec}[e+f x])}{(a+b)\,(c+d \operatorname{Sec}[e+f x])}}}{f\,(b c-a d)\,\sqrt{\frac{c+d}{a+b}}\,\operatorname{Tan}[e+f x]\,\sqrt{-\frac{(b c-a d)\,(1+\operatorname{Sec}[e+f x])}{(a-b)\,(c+d \operatorname{Sec}[e+f x])}}}\operatorname{EllipticE}\left[\operatorname{ArcSin}\left[\sqrt{\frac{c+d}{a+b}}\,\frac{\sqrt{a+b \operatorname{Sec}[e+f x]}}{\sqrt{c+d \operatorname{Sec}[e+f x]}}\right],\,\frac{(a+b)\,(c-d)}{(a-b)\,(c+d)}\right]$$

Program code:

```
Int[sec[e_.+f_.*x_]*(A_+B_.*sec[e_.+f_.*x_])/(Sqrt[a_+b_.*sec[e_.+f_.*x_]]*(c_+d_.*sec[e_.+f_.*x_])^(3/2)),x_Symbol] :=
  2*A*(1+Sec[e+f*x])*Sqrt[(b*c-a*d)*(1-Sec[e+f*x])/((a+b)*(c+d*Sec[e+f*x]))]/
    (f*(b*c-a*d)*Rt[(c+d)/(a+b),2]*Tan[e+f*x]*Sqrt[-(b*c-a*d)*(1+Sec[e+f*x])/((a-b)*(c+d*Sec[e+f*x]))])*
    EllipticE[ArcSin[Rt[(c+d)/(a+b),2]*Sqrt[a+b*Sec[e+f*x]]/Sqrt[c+d*Sec[e+f*x]]],(a+b)*(c-d)/((a-b)*(c+d))] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && EqQ[A,B]
```

```
Int[csc[e_.+f_.*x_]*(A_+B_.*csc[e_.+f_.*x_])/(Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(c_+d_.*csc[e_.+f_.*x_])^(3/2)),x_Symbol] :=
  -2*A*(1+Csc[e+f*x])*Sqrt[(b*c-a*d)*(1-Csc[e+f*x])/((a+b)*(c+d*Csc[e+f*x]))]/
    (f*(b*c-a*d)*Rt[(c+d)/(a+b),2]*Cot[e+f*x]*Sqrt[-(b*c-a*d)*(1+Csc[e+f*x])/((a-b)*(c+d*Csc[e+f*x]))])*
    EllipticE[ArcSin[Rt[(c+d)/(a+b),2]*Sqrt[a+b*Csc[e+f*x]]/Sqrt[c+d*Csc[e+f*x]]],(a+b)*(c-d)/((a-b)*(c+d))] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && EqQ[A,B]
```